

SCIENCE & VIE

MENSUEL

N° 844 JANVIER 1988

EUROCOPTER: LE CHAR VOLANT



**L'HISTOIRE
DU FROMAGE
QUI TUE**

**HI-FI:
BANC
D'ESSAI
DE 21 PAIRES
D'ENCEINTES**

M 2578 - 844 - 17,00 F



3792578017005 08440

SUISSE 8 FS - CANADA \$8,25 - BELGIQUE 120 FB - ESPAGNE 660 Ptas - MAROC 18 Dh - TUNISIE 2,09 DT

Créons des volumes




INFORMATIQUE AMUSANTE

Voici un programme de visualisation en trois dimensions. Très simple, il ne permettra de représenter que des volumes "tournés"; c'est-à-dire des volumes de révolution issus d'un profil tracé sur l'écran. Mais nous pourrions voir la forme ainsi déterminée sous n'importe quel angle en la fai-

l'axe de rotation du profil défini. Ces diverses indications seront affichées par les lignes 220 à 270 pour la ligne d'axe, et par les lignes 280 à 350 pour les commentaires.

Les lignes 410 à 550 du programme seront réservées à la gestion du curseur de détermination du profil. Nous trouverons donc fréquem-

ment l'instruction INKEY qui permettra, d'une part de déplacer le curseur, d'autre part de valider les points du tracé ainsi que le profil lui-même. Vient ensuite la demande de l'angle de rotation pour la visualisation souhaitée du volume créé. Cette fonction sera assurée par les lignes 620 à 660. Notons que la variable P sera modifiée en fonction de l'angle choisi ainsi que la variable AXE contrôlant les coordonnées de la projection de chaque tranche du volume sur son axe vertical. Cette indication four-



sant pivoter dans un plan normal par rapport à celui de l'écran.

Écriture du programme. En premier lieu, nous demanderons à l'Amstrad de travailler en mode 2, afin d'obtenir la meilleure résolution graphique disponible sur l'appareil. Nous initialiserons ensuite, en lignes 70 et 80, les trois variables principales du programme.

P est un facteur correctif destiné à compenser la distorsion du tracé d'un cercle sur l'écran à partir des fonctions sinus et cosinus, XC et YC seront respectivement liées aux positions horizontales et verticales du curseur lors de la détermination du profil.

L'ensemble des rayons ainsi déterminés devra être mémorisé. Pour cela nous créerons un tableau, R(24), à la ligne 140 ainsi que deux tableaux auxiliaires, H(36) et V(36), que nous utiliserons pour le tracé des génératrices du volume.

Ces diverses initialisations et préparations de tableaux effectuées, les données de la forme à représenter pourront être introduites. Pour cela l'écran présentera la liste des touches utilisables pour cette opération, ainsi qu'une ligne de symétrie figurant ici la ligne de

```

10 BORDER 1:CLS:MODE 2
20 REM *****
30 REM *
40 REM *           INITIALISATION DES VARIABLES           *
50 REM *
60 REM *****
70 LET P=0.9
80 LET XC=20:LET YC=1
90 REM *****
100 REM *
110 REM *  CREATION DES TABLEAUX DE MEMORISATION DES DONNEES *
120 REM *
130 REM *****
140 DIM R(24)
150 DIM H(36)
160 DIM V(36)
170 REM *****
180 REM *
190 REM *  MISE EN PLACE DE L'ECRAN DE DETERMINATION DE FORME *
200 REM *
210 REM *****
220 FOR A=20 TO 360 STEP 20
230 PLOT 200,A
240 FOR T=0 TO 20
250 IF T>5 AND T<15 THEN PLOT 200,(A+T)
260 NEXT T
270 NEXT A
280 LOCATE 40,8:PRINT "ENTREZ LES POINTS"
290 LOCATE 40,10:PRINT "CORRESPONDANTS AU"
300 LOCATE 40,12:PRINT "PROFIL DE GENERATION"
310 LOCATE 40,14:PRINT "DE VOTRE VOLUME."
320 LOCATE 40,16:PRINT "POUR DEPLACER LE CURSEUR UTILISEZ"
330 LOCATE 40,18:PRINT "LES TOUCHES PORTANT UNE FLECHE."
340 LOCATE 40,20:PRINT "VALIDEZ VOS POINTS EN TAPANT ENTER."
350 LOCATE 40,22:PRINT "VALIDEZ LE PROFIL EN TAPANT SHIFT-ENTER."
360 REM *****
370 REM *
380 REM *           ENTREE DES DONNEES DU PROFIL           *
390 REM *
400 REM *****
410 LOCATE XC, YC:PRINT " "
420 IF INKEY(1)=0 THEN LET XC=XC+1
430 IF XC>25 THEN LET XC=25
440 IF INKEY(8)=0 THEN LET XC=XC-1
450 IF XC<1 THEN LET XC=1
460 IF INKEY(2)=0 THEN LET YC=YC+1
470 IF YC>24 THEN LET YC=24
480 IF INKEY(0)=0 THEN LET YC=YC-1
490 IF YC<1 THEN LET YC=1
500 LOCATE XC, YC:PRINT "+"
510 IF INKEY(18)=0 THEN LET R(YC)=26-XC:PRINT CHR$(7):LET YC=YC+1
520 IF INKEY(18)=0 THEN GOTO 520

```



nie au programme, la vue du volume, sous l'angle désiré, pourra débiter ; elle sera assurée par les lignes 720 à 840. Elles auront pour fonction d'afficher le numéro de tranche en cours de visualisation ainsi que le nombre de tranches restant à visualiser de même que l'angle de rotation choisi. La boucle FOR... NEXT des lignes 760 à 820 assurera la représentation des 24 tranches du volume. Pour cela nous faisons régulièrement appel au tableau R(24), ainsi qu'à la sous-routine 900 chargée du tracé. En

fin de calcul, la ligne 830 affichera "VISUALISATION TERMINÉE" ; puis le programme sera bouclé de manière à pouvoir demander la visualisation du même volume sous un autre angle.

La sous-routine des lignes 900 à 1020 aura donc pour mission d'assurer l'ensemble du tracé. Il sera effectué par pas de 10 degrés grâce à la boucle FOR... NEXT des lignes 900 à 1000. Ici les fonctions sinus et cosinus seront de nouveau régulièrement utilisées de manière à calculer le tracé de chaque ligne

figurant un cercle vue de biais. Notons également que c'est ici que nous ferons appel aux tableaux H(36) et V(36) pour contrôler le tracé des génératrices du volume.

La frappe ne doit pas poser de problème et l'utilisation de ce programme reste des plus simples. Après l'avoir entièrement tapé, puis demandé RUN, l'écran affichera la ligne d'axe ainsi que les divers commentaires relatifs aux déplacements du curseur. Les flèches "à droite" et "à gauche" seront donc utilisées pour le positionner



```

530 IF INKEY(18)=32 THEN GOTO 610
540 FOR T=1 TO 100:NEXT T
550 GOTO 410
560 REM *****
570 REM *
580 REM *   ENTREE DE L'ANGLE DE ROTATION DE LA FORME CREE   *
590 REM *
600 REM *****
610 CLS
620 IF INKEY$("<>") THEN GOTO 620
630 LOCATE 55,18:PRINT "ANGLE DE ROTATION"
640 LOCATE 55,20:INPUT "EN DEGRES ";AR
650 LET P=SIN ((AR*PI)/180)
660 LET AXE=COS ((AR*PI)/180)
670 REM *****
680 REM *
690 REM *           VISUALISATION DU VOLUME           *
700 REM *
710 REM *****
720 CLS
730 LOCATE 55,2:PRINT "VISUALISATION EN COURS:"
740 LOCATE 55,4:PRINT "ANGLE DE ROTATION:";LOCATE 55,6:PRINT AR;"
  DEGRES."
750 LOCATE 55,8:PRINT "TRANCHE NUMERO:";LOCATE 55,10:PRINT "FIN D
  ANS:   TRANCHES"
760 FOR W=1 TO 24
770 LET RT=8*R(W)
780 LOCATE 71,8:PRINT W
790 LOCATE 64,10:PRINT 24-W
800 LET YR=200+((12-W)*16*AXE)
810 GOSUB 900
820 NEXT W
830 LOCATE 55,2:PRINT "VISUALISATION TERMINEE "
840 GOTO 620
850 REM *****
860 REM *
870 REM *           SOUS ROUTINE DE CALCUL DU TRACE           *
880 REM *
890 REM *****
900 FOR I=10 TO 360 STEP 10
910 LET LT=I/10
920 LET X=200+(RT*SIN ((I*PI)/180))
930 LET XP=200+(RT*SIN (((I+10)*PI)/180))
940 LET Y=YR+((RT*P)*COS(I*PI/180))*0.9
950 LET YP=YR+((RT*P)*COS((I+10)*PI/180))*0.9
960 PLOT X,Y
970 DRAW XP,YP
980 IF W>1 THEN DRAW H(LT),V(LT)
990 LET H(LT)=XP:LET V(LT)=YP
1000 NEXT I
1010 RETURN
1020 LET YR=400-(200+((16*W)-200)*P)

```

et ENTER sera indispensable pour valider chaque point.

Notons que le fait d'appuyer sur ENTER provoque le changement de ligne du curseur. Cependant il sera toujours possible de modifier la position d'un point en utilisant les flèches vers le haut ou vers le bas.

Enfin, une fois le profil entièrement déterminé, les touches SHIFT et ENTER seront pressées simultanément. Dès lors, le programme demandera l'angle de visualisation souhaité. Celui-ci sera indiqué en degrés et validé, encore une fois, par ENTER. La visualisation du volume commencera alors. Notons qu'il sera représenté en perspective cavalière et comme s'il était constitué d'une matière transparente.

Certes, ce programme n'est qu'un embryon de représentation en trois dimensions, notamment parce qu'il ne permet de représenter que des formes de révolution et n'effectue leur rotation que dans un seul plan. Néanmoins, il pourra faire office de base à bien d'autres applications dont, nous n'en doutons pas, bon nombre d'entre vous souhaiteront entreprendre la mise au point.

Henri-Pierre Penel